EASTERN UNIVERSITY

Department of Computer Science and Engineering (CSE)

Microcontroller and Interfacing Lab

Project Title :- Multi-Layered Smart Home Security and Automation System using ESP32.

Submitted By:

1. **Shawon Gazi**
   241400031
2. **Tamanna Ahmed Tasin**
   241400033
3. **Md. Al-Sharif**
   241400037

Course Teacher:

Tazeen Tasneem

Assistant Professor, Department of **CSE**

Eastern University

PICTURE OF OUR DESIGN



Component Substitution For Online Simulation

Due to limitations of the Wokwi platform, the following substitutions were made:

Original Component	**Availability in Wokwi**	**Alternative Used**	**Reason**
IR Sensor	Not available	Push Button	To simulate object detection (beam break)
Laser Diode + LDR	Not available	Push Button	To simulate intrusion detection
Relay Module	Not available	LED	To simulate ON/OFF switching
Logic Level Converter	Not required	Not used	Simulation does not require voltage conversion

These substitutions allowed the system logic to be tested effectively without affecting the overall functionality.

CODE OF OUR DESIGN

```
#include <Wire.h>
#include <Keypad.h>
#include <ESP32Servo.h>
#include <LiquidCrystal_I2C.h>

// ================== LCD ==================
#define LCD_ADDR 0x27
LiquidCrystal_I2C lcd(LCD_ADDR, 16, 2);

// ================== PIN CONFIG ==================
#define IR_START 12
#define IR_END 13
#define LASER_PIN 26
#define BUZZER_PIN 25
#define TRIG_PIN 14
#define ECHO_PIN 27
#define RELAY_PIN 33
#define GATE_SERVO_PIN 18
#define DOOR_SERVO_PIN 19

// ================== SETTINGS ==================
const char masterPin[] = "1234";
float speedLimit = 20.0;
float distanceBetweenIR = 10.0;

// ================== SERVO ==================
Servo gateServo;
Servo doorServo;

// ================== KEYPAD ==================
const byte ROWS = 4;
const byte COLS = 4;

char keys[ROWS][COLS] = {
  {'1','2','3','A'},
  {'4','5','6','B'},
  {'7','8','9','C'},
  {'*','0','#','D'}
};
```

```
byte rowPins[ROWS] = {32, 35, 34, 39};
byte colPins[COLS] = {23, 5, 17, 16};

Keypad keypad = Keypad(makeKeymap(keys), rowPins, colPins, ROWS, COLS);

// ================== VARIABLES ==================
String inputPin = "";
int wrongAttempts = 0;

unsigned long startTime = 0;
bool timing = false;

unsigned long lastServoAction = 0;
bool gateOpen = false;
bool doorOpen = false;

// ================== SETUP ==================
void setup() {
  Wire.begin();

  lcd.init();
  lcd.backlight();

  pinMode(IR_START, INPUT);
  pinMode(IR_END, INPUT);
  pinMode(LASER_PIN, INPUT);
  pinMode(BUZZER_PIN, OUTPUT);
  pinMode(RELAY_PIN, OUTPUT);
  pinMode(TRIG_PIN, OUTPUT);
  pinMode(ECHO_PIN, INPUT);

  gateServo.attach(GATE_SERVO_PIN);
  doorServo.attach(DOOR_SERVO_PIN);

  gateServo.write(0);
  doorServo.write(0);

  lcd.setCursor(0,0);
  lcd.print("Smart Home");
  lcd.setCursor(0,1);
  lcd.print("System Ready");
  delay(1500);
```

```
  lcd.clear();
}

// ================== MAIN LOOP ==================
void loop() {

  //  1. LASER SECURITY (Highest Priority)
  if (digitalRead(LASER_PIN) == LOW) {
    triggerAlarm("SECURITY ALERT!", "Intruder!");
    return;
  }

  //  2. SPEED CHECK
  checkSpeed();

  //  3. KEYPAD
  handleKeypad();

  //  4. AUTO LIGHT
  checkRoomLight();

  // ⏱ 5. AUTO CLOSE SERVOS (Non-blocking)
  handleServoAutoClose();
}

// ================== SPEED FUNCTION ==================
void checkSpeed() {
  if (digitalRead(IR_START) == LOW && !timing) {
    startTime = millis();
    timing = true;
  }

  if (digitalRead(IR_END) == LOW && timing) {
    unsigned long duration = millis() - startTime;
    float speed = distanceBetweenIR / (duration / 1000.0);
    timing = false;

    lcd.clear();
    lcd.print("Speed:");
    lcd.print(speed);

    if (speed > speedLimit) {
      lcd.setCursor(0,1);
      lcd.print("Too Fast!");
```

```
      digitalWrite(BUZZER_PIN, HIGH);
      delay(1500);
      digitalWrite(BUZZER_PIN, LOW);
    } else {
      lcd.setCursor(0,1);
      lcd.print("Gate Opening");

      gateServo.write(90);
      gateOpen = true;
      lastServoAction = millis();
    }
    delay(1000);
    lcd.clear();
  }
}

// ================== KEYPAD FUNCTION ==================
void handleKeypad() {
  char key = keypad.getKey();

  if (!key) return;

  if (key == '#') {
    if (inputPin == masterPin) {
      lcd.clear();
      lcd.print("Access Granted");

      doorServo.write(90);
      doorOpen = true;
      lastServoAction = millis();

      wrongAttempts = 0;
    } else {
      wrongAttempts++;
      lcd.clear();
      lcd.print("Wrong PIN");

      if (wrongAttempts >= 3) {
        triggerAlarm("LOCKED!", "3 Wrong PIN");
        wrongAttempts = 0;
      }
    }
    inputPin = "";
  }
```

```

  else if (key == '*') {
    inputPin = "";
    lcd.clear();
  }

  else {
    inputPin += key;
    lcd.setCursor(0,1);
    lcd.print("PIN: ****");
  }
}

// ================== AUTO LIGHT ==================
void checkRoomLight() {

  digitalWrite(TRIG_PIN, LOW);
  delayMicroseconds(2);

  digitalWrite(TRIG_PIN, HIGH);
  delayMicroseconds(10);

  digitalWrite(TRIG_PIN, LOW);

  long duration = pulseIn(ECHO_PIN, HIGH);
  int distance = duration * 0.034 / 2;

  if (distance > 0 && distance < 50) {
    digitalWrite(RELAY_PIN, HIGH);
  } else {
    digitalWrite(RELAY_PIN, LOW);
  }
}

// ================== SERVO AUTO CLOSE ==================
void handleServoAutoClose() {

  if (gateOpen && millis() - lastServoAction > 5000) {
    gateServo.write(0);
    gateOpen = false;
  }

  if (doorOpen && millis() - lastServoAction > 5000) {
    doorServo.write(0);
```

```
    doorOpen = false;
  }
}

// ================== ALARM ===================
void triggerAlarm(String msg1, String msg2) {
  lcd.clear();
  lcd.setCursor(0,0);
  lcd.print(msg1);
  lcd.setCursor(0,1);
  lcd.print(msg2);

  digitalWrite(BUZZER_PIN, HIGH);
  delay(2000);
  digitalWrite(BUZZER_PIN, LOW);

  lcd.clear();
}
```